

# ORACLE 12c

**PL/SQL – Обработка исключений**

**Лекция 11**

# Обработка исключений


---

- ▶ Исключительная ситуация – событие, возникающее в программе и требующее незамедлительной обработки
- ▶ Два типа исключительных ситуаций:
  - ▶ 1) программно-определяемые исключения;
  - ▶ 2) предопределенные (стандартные) исключения.



# Обработка исключений

- ▶ Ошибка, сгенерированная сервером
- ▶ Ошибка в результате действий пользователя
- ▶ Ошибка, сгенерированная приложением пользователю



# Стратегия обработки исключений

---

- ▶ Как и где будут фиксироваться ошибки, чтобы их можно было просмотреть и откорректировать?
- ▶ Как выдавать пользователю сообщения об ошибках?
- ▶ Нужно ли включать обработку исключений в каждый PL/SQL блок?
- ▶ Как управлять транзакцией в случае ошибки?



# Термины обработки исключений

---

- ▶ Секция исключений – необязательная секция в PL/SQL блоке, которая содержит один или несколько обработчиков исключений
- ▶ RAISE (RAISE\_APPLICATION\_ERROR) – команда, которая прерывает выполнение текущего блока
- ▶ Обработка исключений – перехват ошибки в секции исключений
- ▶ Область действия – часть кода, в рамках которого может быть сгенерировано исключение
- ▶ Распространение исключения – процесс передачи исключений от одного блока другому, если исключение не было обработано



# Термины обработки исключений

---


- ▶ Необработанное исключение – исключение становится необработанным, если оно не обработано блоком самого верхнего уровня
- ▶ Неименованное исключение – исключение, которое имеет код ошибки и сообщение, но не имеет наименования, не может быть использовано в команде RAISE или в секции WHEN
- ▶ Именованное исключение – исключение, которому было определено наименование



# Предопределенные исключения

```
-- 14/05/2014
declare
begin
null;
-- DUP_VAL_ON_INDEX          нарушена уникальность
-- TIMEOUT_ON_RESOURCE       истекло время ожидания ресурса
-- TRANSACTION_BACKED_OUT
-- INVALID_CURSOR            запрещенная операция с курсором
-- NOT_LOGGED_ON              отсутствует подключение к Oracle
-- NO_DATA_FOUND              данные не найдены
-- SYS_INVALID_ROWID         не может быть получен ROWID
-- TOO_MANY_ROWS              не точная выборка в SELECT
-- ZERO_DIVIDE                деление на ноль
-- USERENV_COMMITSCN_ERROR   ошибка USERENV('COMMITSCN')
-- INVALID_NUMBER             ошибка преобразования в NUMBER
-- STORAGE_ERROR              PL/SQL недостаточно памяти
-- PROGRAMM_ERROR             внутренняя ошибка PL/SQL
-- VALUE_ERROR                ошибка преобразование или усечение точности
-- ROWTYPE_MISMATCH           переменная PL/SQL и курсорная переменная не совместимы
-- CURSOR_ALREADY_OPEN        попытка открыть открытый курсор
-- ACCESS_INTO_NULL           попытка присвоить значение атрибуту NULL-объекта
-- CASE_NOT_FOUND             нет подходящей фразы WHEN в операторе CASE
-- SELF_IS_NULL               попытка вызвать метод NULL-объекта
-- INVALID_PATH
-- INVALID_MODE
-- INVALID_FILEHANDLE
-- INVALID_OPERATION
-- READ_ERROR
-- WRITE_ERROR
-- INTERNAL_ERROR
-- INVALID_MAXLINESIZE
-- INVALID_FILENAME
-- ACCESS_DENIED
-- INVALID_OFFSET
-- DELETE_FAILED
-- RENAME_FAILED
-- NO_DATA_NEEDED
-- collection_exceptions      см. типы исключений для коллекций
end;
```

# Sqlerrm и sqlcode



The screenshot displays the Oracle SQL Developer environment. The top toolbar includes icons for file operations, execution, and formatting, along with a timer showing 0,359 seconds. The main editor window contains a SQL script named 14/11.sql. The script defines a procedure that outputs various Oracle error codes using the sqlerrm function. The output window, titled 'Dbms Output', shows the results of the script execution, including a list of error codes and their descriptions. The error codes are: ORA-0000: normal, successful completion; User-Defined Exception; ORA-01403: данные не найдены; ORA-00001: нарушено ограничение уникальности (.); ORA-06502: PL/SQL: ошибка числа или значения; ORA-19999: была вызвана процедура skip\_row; ORA-20000: (highlighted in red); ORA-20001: (highlighted in red); ORA-20800: (highlighted in red); ORA-20999: (highlighted in red); ORA-21000: номер ошибки для raise\_application\_error вне диапазона; ORA-29001: UNUSED; and ORA-29002: Транспорт SSL обнаружил недопустимый или устаревший серт.

```
-- 14/11.sql
declare
begin
  dbms_output.put_line(sqlerrm(0));
  dbms_output.put_line(sqlerrm(1));
  dbms_output.put_line(sqlerrm(100));
  dbms_output.put_line(sqlerrm(-1));
  dbms_output.put_line(sqlerrm(-6502));
  dbms_output.put_line(sqlerrm(-19999));
  dbms_output.put_line(sqlerrm(-20000));
  dbms_output.put_line(sqlerrm(-20001));
  dbms_output.put_line(sqlerrm(-20800));
  dbms_output.put_line(sqlerrm(-20999));
  dbms_output.put_line(sqlerrm(-21000));
  dbms_output.put_line(sqlerrm(-29001));
  dbms_output.put_line(sqlerrm(-29002));
end;
```

Dbms Output x

Buffer Size: 20000 Poll

shl-svvcore x

ORA-0000: normal, successful completion  
User-Defined Exception  
ORA-01403: данные не найдены  
ORA-00001: нарушено ограничение уникальности (.)  
ORA-06502: PL/SQL: ошибка числа или значения  
ORA-19999: была вызвана процедура skip\_row  
ORA-20000:  
ORA-20001:  
ORA-20800:  
ORA-20999:  
ORA-21000: номер ошибки для raise\_application\_error вне диапазона  
ORA-29001: UNUSED  
ORA-29002: Транспорт SSL обнаружил недопустимый или устаревший серт



# Объявление именованных исключений

---

- ▶ Чтобы обработать исключение, которое не относится к определенным сервером, его необходимо объявить:
  - ▶ `exception_name EXCEPTION;`
- ▶ Имена исключений могут быть использованы только для генерации исключения при помощи `RAISE` и для перехвата исключения в секции `WHEN`



# Связывание исключений с кодом ошибки

---

- ▶ Синтаксис:


- ▶ `exception_name EXCEPTION;`
- ▶ `PRAGMA EXCEPTION_INIT (exception_name, integer);`

- ▶ Где `exception_name` – наименование исключения, `integer` – номер(код) ошибки сервера ORACLE, с которым необходимо связать исключение




# Генерация и обработка исключений

```
-- 14/06.sql
declare
    e_namesake exception;
    n NUMBER(5);
begin
    select count(*) into n
    from teacher t1, teacher t2
    where t1.teacher_name = t2.teacher_name and
          t1.teacher != t2.teacher;
    if n = 0 then raise e_namesake;
    end if;
    dbms_output.put_line(n);
exception
    when no_data_found
    then dbms_output.put_line('no_data_found');
    when not_logged_on
    then dbms_output.put_line('no_logged_on');
    when timeout_on_resource
    then dbms_output.put_line('timeout_on_resource');
    when others
    then dbms_output.put_line('sqlerrm = '||sqlerrm);
    dbms_output.put_line('sqlcode = '||sqlcode);
end;
```



# Генерация и обработка исключений

```
-- 14/07.sql
declare
  e_namesake exception;
  n NUMBER(5);
begin
  select count(*) into n
  from teacher t1, teacher t2
  where t1.teacher_name = t2.teacher_name and
        t1.teacher != t2.teacher;
  if n = 0 then raise e_namesake;
  end if;
  dbms_output.put_line(n);
exception
  when no_data_found
  then dbms_output.put_line('no_data_found');
  when not_logged_on
  then dbms_output.put_line('no_logged_on');
  when timeout_on_resource
  then dbms_output.put_line('timeout_on_resource');
  when others
  then dbms_output.put_line('sqlerrm = '||sqlerrm);
    dbms_output.put_line('sqlcode = '||sqlcode);
    dbms_output.put_line(
      substr(dbms_utility.format_error_stack, 1,200)
    );
    dbms_output.put_line(
      substr(dbms_utility.format_error_backtrace, 1,200)
    );
end;
```



# Генерация и обработка исключений

-- 14/08.sql

**declare**

**e\_namesake exception;**

**n** **NUMBER**(5);

**begin**

**select** **count**(\*) **into** **n**

**from** **teacher** **t1**, **teacher** **t2**

**where** **t1.teacher\_name** = **t2.teacher\_name** **and**  
**t1.teacher** != **t2.teacher**;

**if** **n** = 0 **then** **raise** **e\_namesake**;

**end if**;

**dbms\_output.put\_line**(**n**);

**exception**

**when** **no\_data\_found**

**then** **dbms\_output.put\_line**('no\_data\_found');

**when** **not\_logged\_on**

**then** **dbms\_output.put\_line**('no\_logged\_on');

**when** **timeout\_on\_resource**

**then** **dbms\_output.put\_line**('timeout\_on\_resource');

**when** **e\_namesake**


**then** **dbms\_output.put\_line**('e\_namesake');

**when** **others**


**then** **dbms\_output.put\_line**('sqlerrm = '||**sqlerrm**);

**dbms\_output.put\_line**('sqlcode = '||**sqlcode**);

**end;**



# Генерация и обработка исключений



The screenshot displays the SQL Worksheet interface. The main editor shows an SQL script with line numbers 1 through 26. The script declares an exception `e_namesake`, initializes it with `pragma exception_init(e_namesake, 100);`, and then uses a `select` statement to find duplicate teacher names. An `exception` block handles various errors, including `when e_namesake` which outputs `'e_namesake'`. A `Dbms Output` window is open on the right, showing the output of the script. The bottom status bar indicates 'Task completed in 0,046 seconds'.

```
1  -- 14/09.sql
2  declare
3      e_namesake exception;
4      pragma exception_init(e_namesake, 100);
5      t teacher.teacher_name%type;
6
7  begin
8      select t1.teacher into t
9      from teacher t1, teacher t2
10     where t1.teacher_name = t2.teacher_name and
11           t1.teacher != t2.teacher and
12           rownum < 2;
13
14     dbms_output.put_line(t);
15
16 exception
17     when no_data_found
18     then dbms_output.put_line('no_data_found');
19     when not_logged_on
20     then dbms_output.put_line('no_logged_on');
21     when timeout_on_resource
22     then dbms_output.put_line('timeout_on_resource');
23     when e_namesake
24     then dbms_output.put_line('e_namesake');
25     when others
26     then dbms_output.put_line('sqlerrm = '||sqlerrm);
27         dbms_output.put_line('sqlcode = '||sqlcode);
28
29 end;
```

Dbms Output x

shl-svvcore x

Task completed in 0,046 seconds

Error report:


ORA-06550: Строка 19, столбец 8:

PLS-00484: дополнительные исключения 'NO DATA FOUND' и 'E NAMESAKE' должны появляться в том же обработчике исключений

ORA-06550: Строка 0, столбец 0:

# Генерация и обработка исключений


```
-- 14/10.sql
declare
    e_namesake exception;
    pragma exception_init(e_namesake, 100);
    t teacher.teacher_name%type;
begin
    select t1.teacher into t
    from teacher t1, teacher t2
    where t1.teacher_name = t2.teacher_name and
          t1.teacher != t2.teacher and
          rownum < 2;
    dbms_output.put_line(t);
exception
    -- when no_data_found
    -- then dbms_output.put_line('no_data_found');
    when not_logged_on
    then dbms_output.put_line('no_logged_on');
    when timeout_on_resource
    then dbms_output.put_line('timeout_on_resource');
    when e_namesake
    then dbms_output.put_line('e_namesake');
    when others
    then dbms_output.put_line('sqlerrm = '||sqlerrm);
        dbms_output.put_line('sqlcode = '||sqlcode);
end;
```



# RAISE\_APPLICATION\_ERROR

-- 14/12.sql

```
declare
    e_namesake exception;
    pragma exception_init(e_namesake, 100);
    t teacher.teacher_name%type;
begin
    select t1.teacher into t
    from teacher t1, teacher t2
    where t1.teacher_name = t2.teacher_name and
          t1.teacher != t2.teacher and
          rownum < 2;
    dbms_output.put_line(t);
exception
    -- when no_data_found
    when not_logged_on
    then dbms_output.put_line('no_logged_on');
    when timeout_on_resource
    then dbms_output.put_line('timeout_on_resource');
    when e_namesake
    then dbms_output.put_line('e_namesake');
        raise application_error(-20001, 'e_namesake');
    when others
    then dbms_output.put_line('sqlerrm = '||sqlerrm);
        dbms_output.put_line('sqlcode = '||sqlcode);
end;
```



Error report:

ORA-20001: e namesake

ORA-06512: ha line 20



# RAISE\_APPLICATION\_ERROR


---

- ▶ Определена в пакете DBMS\_STANDARD
- ▶ Можно присвоить сообщение об ошибке
- ▶ При выполнении процедуры:
  - ▶ Выполнение блока прерывается
  - ▶ Любые изменения в аргументах IN OUT и OUT откатываются
  - ▶ Изменения в глобальных структурах (пакетные переменные, объекты базы данных) не откатываются – для отката надо явно выполнить ROLLBACK




# Распространение исключения

```
-- 14/13.sql
declare
    aa exception;
begin          -- 1
    begin      -- 2
        dbms_output.put_line('block-2');
        raise aa;
    exception
        when aa then dbms_output.put_line('block-2:aa');
    end;        -- 2
    dbms_output.put_line('block-1');
exception
    when others then dbms_output.put_line('block-1:others');
end;           -- 1
```




# Распространение исключения

```
-- 14/14.sql
declare
    aa exception;
    bb exception;
begin          -- 1
    begin      -- 2
        dbms_output.put_line('block-2');
        raise bb;
    exception
        when aa then dbms_output.put_line('block-2:aa');
    end;        -- 2
    dbms_output.put_line('block-1');
exception
    when bb then dbms_output.put_line('block-1:bb');
    when others then dbms_output.put_line('block-1:others');
end;           -- 1
```




# Распространение исключения

```
17/10/2014  
  
declare  
    aa exception;  
    bb exception;  
    cc exception;  
begin      -- 1  
    begin  -- 2  
        dbms_output.put_line('block-2');  
        raise aa;  
    exception  
        when aa then dbms_output.put_line('block-2:aa');  
            raise bb;  
        when others then  
            dbms_output.put_line('block-2:other');  
    end;      -- 2  
    dbms_output.put_line('block-1');  
exception  
    when bb then  
        dbms_output.put_line('block-1:bb');  
  
end;      -- 1
```



# Распространение исключения

```
-- 14/16.sql
declare
  aa exception;
  bb exception;
  cc exception;
begin
  -- 1
  begin
    -- 2
    dbms_output.put_line('block-2');
    raise cc;
  exception
    when aa then dbms_output.put_line('block-2:aa');
  end;
  -- 2
  dbms_output.put_line('block-1');
exception
  when bb then dbms_output.put_line('block-1:bb');
  -- when others then dbms_output.put_line('block-1:others');
end;
-- 1
```



Error report:

ORA-06510: PL/SQL: необработанное исключение, определенное пользователем

ORA-06512: на line 8

ORA-06510: необработанное исключение, определенное пользователем

# Вопросы?

---

